
اتصال خودکار و مجدد به مودم پس از قطع ارتباط در ماژول
ESP8266

تاریخ انتشار۲۲ تیر, ۱۴۰۲ توسط   سید حسین سلطان

ESP8266 از سری مطالب مربوط به آموزش های ام. در مطالب قبلروالمای سلام و درود خدمت همراهان همیش
به ارسال و دریافت دیتا از طریق MQTT پرداخته شد. در این مطلب به بررس و نحوه اتصال مجدد بصورت خودکار
به آن متصل بوده است پرداخته خواهد شد. پس با من تا انتهای مطلب  ESP8266 به مودم یا روتری که ماژول وایفای

همراه باشید. همچنین شما میتونید سایر مطالب من رو از این قسمت دنبال کنید.

https://blog.microele.com/category/esp8266/
https://blog.microele.com/category/esp8266/
https://blog.microele.com/send-receive-data-over-mqtt-with-esp8266/
https://blog.microele.com/send-receive-data-over-mqtt-with-esp8266/
https://blog.microele.com/author/soltanimicroele-com/

مقدمه
برخ از پروژه های مبتن بر اینترنت اشیا (IoT) که از ماژول های ESP8266 یا ESP32 استفاده شده است مستلزم اتصال
ESP8266 ماژول ار طبق آموزش های قبلمودم یا روتر جهت برقراری ارتباط بر بستر اینترنت هستند. برای این به ی
یا ESP32 خود را به آن متصل مکنیم. اما در برخ مواقع بنا به دلایل همانند Restart شدن مودم، قطع اتصال اینترنت

مودم، خاموش شدن مودم یا هر دلیل دیر ارتباط دستاه با مودم قطع خواهد شد. 

در این مواقع نیاز است که دستاه به مودم دیری متصل شود یا در صورت رفع مواردی که بیان شد، مجددا به همان
های موجود بررس شود و در صورت  WiFi هوجود دارد مثلا دائما شب ه متصل شود. در این مورد روش هایشب
وجود شبه مورد نظر مجددا به آن متصل شد. اما ی روش ساده تر این است که بصورت خودکار ماژول به آخرین

یا همان مودم/روتر مد نظر متصل شود. در این مطلب به بیان این راه حل پرداخته خواهد شد.

WiFi به مودم ESP8266 اتصال
به مودم یا روتر متوانیم از ته کد زیر استفاده نماییم. در کد زیر پس از  EPS8266 جهت برقراری ارتباط و اتصال 
اتصال موفقیت آمیز، IP محل اختصاص داده شده را نمایش خواهیم داد. در کد زیر بجای ssid نام مودم خود و بجای
رمز مودم را وارد کنید. حتما به یاد داشته باشید که این دو مقدار را حتما بین دو دبل کوتیشن (" ") قرار  ،password

دهید.

void initWiFi() {

  WiFi.mode(WIFI_STA);

  WiFi.begin(ssid, password);

  Serial.print("Connecting to WiFi ..");

  while (WiFi.status() != WL_CONNECTED) {

    Serial.print('.');

    delay(1000);

  }

  Serial.println(WiFi.localIP());

}

ESP8266 پس از قطع اتصال در WiFi هاتصال مجدد به شب
اتصال مجدد بصورت خودکار به مودم یا روتر خواهد بود. مثلا دستاه به مودم متصل  ESP8266 از قابلیت های ی
شده است اما به هر دلیل اتصال قطع خواهد شد. حال پس از روشن شدن مجدد، دستاه باید به همین مودم متصل

شود. با استفاده از این آموزش متوان اینار را براحت انجام دهیم.

پس از قطع ارتباط متوانیم از دستورات به ترتیب زیر استفاده نماییم. با این کار  WiFi جهت اتصال مجدد به شبه 
متوانیم ماژول ESP8266 را به آخرین شبه ای که متصل بوده است، متصل نماییم. محل قرار گیری این دستورات، در
جای است که دستورات برقراری ارتباط و اتصال به مودم را نوشته ایم. متوانیم در کد های که در بالا به آن اشاره

شده، آنها را پس از دستور Serial.println(WiFi.localIP()) وارد نماییم.

WiFi.setAutoReconnect(true);

WiFi.persistent(true);

ESP8266 در WiFi هبرقراری ارتباط با مودم و اتصال خودکار به شب
در کد های زیر، هر 20 ثانیه وضعیت اتصال به مودم بررس خواهد شد. در صورت که مشل رخ داده باشد مثلا
مودم قطع یا خاموش شده باشد وضعیت آن را برروی سریال مانیتور نمایش خواهیم داد. همچنین با استفاده از
دستورات مربوط به اتصال مجدد که در کد زیر نیز لحاظ شده است، پس از روشن شدن یا رفع مشل مودم، ماژول
استفاده شده  Millis بصورت خودکار به مودم متصل خواهد شد. جهت بررس به مدت هر 20 ثانیه از تابع   ESP8266

است.

#include <ESP8266WiFi.h>

const char* ssid = "ssid";

const char* password = "password";

unsigned long previousMillis = 0;

unsigned long interval = 20000; //every 20sec.

void initWiFi() {

  WiFi.mode(WIFI_STA);

  WiFi.begin(ssid, password);

  Serial.print("Connecting to WiFi ..");

  while (WiFi.status() != WL_CONNECTED) {

    Serial.print('.');

    delay(1000);

  }

  Serial.println("");

  Serial.print("Local IP: ");

  Serial.println(WiFi.localIP());

  WiFi.setAutoReconnect(true);

  WiFi.persistent(true);

}

void setup() {

  Serial.begin(115200);

  initWiFi(); //connect to modem

}

void loop() {

  unsigned long currentMillis = millis();

  if (currentMillis - previousMillis >= interval){

    //check ESP8266 WiFi connection status

    switch (WiFi.status()){

      case WL_NO_SSID_AVAIL:

        Serial.println("Configured SSID cannot be reached");

        break;

      case WL_CONNECTED:

        Serial.println("Connection successfully established");

        break;

      case WL_CONNECT_FAILED:

        Serial.println("Connection failed");

        break;

    }

    Serial.printf("Connection status: %d\n", WiFi.status());

    previousMillis = currentMillis;

  }

}

نتیجه

ESP8266 اتصال مجدد به مودم پس از قطع ارتباط در

ESP8266 در Events اتصال مجدد به مودم به روش استفاده از وقایع یا
توسط ماژول های ESP8266 متوانیم Event ها گوناگون را همانند قطع ارتباط داشته و از آنها در کد های خود استفاده
نماییم. با استفاده از این وقایع نیاز به چ کردن وضعیت ماژول بصورت دائم نیست. هنام که یEvent  رخ دهد،
تابع معرف شده آن و دستورات کاربر در آن، بصورت خودکار فراخوان و انجام خواهند شد. متوان این کار را به
نوع شبیه عملرد وقفه دانست. در روش دوم اتصال مجدد ماژول به مودم یا روتر از Event های زیر متوانیم استفاده

نماییم. در ادامه به ی نمونه از استفاده این دستورات پرداخته خواهد شد.

اختصاص  IP توانیمام اتصال بصورت موفقیت آمیز، این رویداد رخ داده و مهن  :onStationModeGotIP
داده شده به ماژول را دریافت و آن را نمایش دهیم.

هنام قطع ارتباط با مودم، این رویداد رخ داده و متوانیم دستورات مورد  :onStationModeDisconnected
نظر خود را در صورت رخ دادن چنین وضعیت انجام دهیم.

نمونه کد
لوکال اختصاص  IP رخ داده و در تابع آن  (onWifiConnect) مورد نظر  Event ،ام اتصال صحیح به مودمدر کد زیر هن
داده شده را نمایش خواهیم داد. در صورت قطع ارتباط ماژول و مودم Event مربوطه (onWifiDisconnect) رخ داده و در

دستورات داخل تابع آن مجددا نسبت به اتصال به مودم اقدام خواهد شد.

#include <ESP8266WiFi.h>

// Replace with your network credentials

const char* ssid = "ssid";

const char* password = "password";

WiFiEventHandler wifiConnectHandler;

WiFiEventHandler wifiDisconnectHandler;

void initWiFi() {

  WiFi.mode(WIFI_STA);

  WiFi.begin(ssid, password);

  Serial.print("Connecting to WiFi ..");

  while (WiFi.status() != WL_CONNECTED) {

    Serial.print('.');

    delay(1000);

  }

  Serial.println("");

  Serial.print("Local IP: ");

  Serial.println(WiFi.localIP());

}

void onWifiConnect(const WiFiEventStationModeGotIP& event) {

  Serial.println("Connected to Wi-Fi sucessfully.");

  Serial.print(" Local IP Address: ");

  Serial.println(WiFi.localIP());

}

void onWifiDisconnect(const WiFiEventStationModeDisconnected& event) {

  Serial.println("Disconnected from Wi-Fi, trying to connect...");

  WiFi.disconnect();

  WiFi.begin(ssid, password);

}

void setup() {

  Serial.begin(115200);

  pinMode(2, OUTPUT);

  //Register event handlers

  wifiConnectHandler = WiFi.onStationModeGotIP(onWifiConnect);

  wifiDisconnectHandler = WiFi.onStationModeDisconnected(onWifiDisconnect);

  //connect to modem

  initWiFi();

}

void loop() {

  digitalWrite(2, LOW);

  delay(1000);

  digitalWrite(2, HIGH);

  delay(1000);

}

نتیجه

ESP8266 در Events اتصال مجدد به مودم پس از قطع ارتباط مودم با استفاده از

نتیجه گیری
در این مطلب به بیان دو روش جهت برقرای ارتباط و اتصال مجدد ماژول ESP8266 با مودم یا روتر وایفای بیان شد. با
استفاده از روش دوم یا همان استفاده از Event ها نیاز به بررس همیش و دائم نبوده و در صورت رخ دادن شرایط

Event، به تابع مربوطه رفته و متوانیم دستورات مورد نظر خود را انجام دهیم.

امیدوارم از این آموزش کمال بهره را برده باشید. در صورت که هرگونه نظر یا سوال داشتید درباره این آموزش لطفا
اون رو در انتهای همین صفحه در قسمت دیدگاه ها قرار بدید. در کوتاه ترین زمان ممن به اون ها پاسخ خواهم داد.
اگر این مطلب براتون مفید بود، اون رو حتما به اشتراک بذارید. همینطور میتونید این آموزش را توی اینستاگرام با

هشتmicroelecom # به اشتراک بذارید و پیج مایروالام (microelecom@) رو هم منشن کنید.

